In [ ]:
```julia
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import FiniteDiff
import ForwardDiff as FD
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
using MeshCat
const mc = MeshCat
using StaticArrays
using Printf
```

```
 Activating project at `~/Course/CMU16-745-Optimal-Control-HW/hw4`
```

In [ ]:
```julia
include(joinpath(@__DIR__, "utils","ilc_visualizer.jl"))
```

```
update_car_pose! (generic function with 1 method)
```

Q1: Iterative Learning Control (ILC) (40 pts)

In this problem, you will use ILC to generate a control trajectory for a Car as it swerves to avoid a moose, also known as "the moose test" (wikipedia, video). We will model the dynamics of the car as with a simple nonlinear bicycle model, with the following state and control:

$$\begin{align} x = \begin{bmatrix} p_x \\ p_y \\ \theta \\ \delta \\ v \end{bmatrix}, &\quad \quad u = \begin{bmatrix} a \\ \dot{\delta} \end{bmatrix} \end{align} $$

where p_x and p_y describe the 2d position of the bike, θ is the orientation, δ is the steering angle, and v is the velocity. The controls for the bike are acceleration a, and steering angle rate $\dot{\delta}$.

In [ ]:
```julia
function estimated_car_dynamics(model::NamedTuple, x::Vector, u::Vector)::Vector
    # nonlinear bicycle model continuous time dynamics
    px, py, θ, δ, v = x
    a, δdot = u

    β = atan(model.lr * δ, model.L)
    s,c = sincos(θ + β)
    ω = v*cos(β)*tan(δ) / model.L

    vx = v*c
    vy = v*s

    xdot = [
        vx,
        vy,
        ω,
        δdot,
        a
    ]

    return xdot
```

```
end
function rk4(model::NamedTuple, ode::Function, x::Vector, u::Vector, dt::Real)::Vector
    k1 = dt * ode(model, x,        u)
    k2 = dt * ode(model, x + k1/2, u)
    k3 = dt * ode(model, x + k2/2, u)
    k4 = dt * ode(model, x + k3,   u)
    return x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end
```

```
rk4 (generic function with 1 method)
```

We have computed an optimal trajectory X_{ref} and U_{ref} for a moose test trajectory offline using this `estimated_car_dynamics` function. Unfortunately, this is a highly approximate dynamics model, and when we run U_{ref} on the car, we get a very different trajectory than we expect. This is caused by a significant sim to real gap. Here we will show what happens when we run these controls on the true dynamics:

In [ ]:

```
function load_car_trajectory()
    # load in trajectory we computed offline
    path = joinpath(@__DIR__, "utils","init_control_car_ilc.jld2")
    F = jldopen(path)
    Xref = F["X"]
    Uref = F["U"]
    close(F)
    return Xref, Uref
end
function true_car_dynamics(model::NamedTuple, x::Vector, u::Vector)::Vector
    # true car dynamics
    px, py, θ, δ, v = x
    a, δdot = u

    # sluggish controls (not in the approximate version)
    a = 0.9*a - 0.1
    δdot = 0.9*δdot - .1*δ + .1

    β = atan(model.lr * δ, model.L)
    s,c = sincos(θ + β)
    ω = v*cos(β)*tan(δ) / model.L

    vx = v*c
    vy = v*s

    xdot = [
        vx,
        vy,
        ω,
        δdot,
        a
    ]

    return xdot
end

@testset "sim to real gap" begin
    # problem size
    nx = 5
    nu = 2
    dt = 0.1
    tf = 5.0
    t_vec = 0:dt:tf
```

```julia
    N = length(t_vec)
    model = (L = 2.8, lr = 1.6)

    # optimal trajectory computed offline with approximate model
    Xref, Uref = load_car_trajectory()

    # TODO: simulated Uref with the true car dynamics and store the states in Xsim
    Xsim = [Xref[1]]
    for i in 1:N-1
        x = Xsim[end]
        u = Uref[i]
        xnext = rk4(model, true_car_dynamics, x, u, dt)
        push!(Xsim, xnext)
    end


    # --------testing-----------------
    @test norm(Xsim[1] - Xref[1]) == 0
    @test norm(Xsim[end] - [3.26801052, 15.0590156, 2.0482790, 0.39056168, 4.5],Inf) < 1e-4

    # --------plotting/animation-------
    Xm= hcat(Xsim...)
    Xrefm = hcat(Xref...)
    plot(Xrefm[1,:], Xrefm[2,:], ls = :dash, label = "reference",
         xlabel = "x (m)", ylabel = "y (m)", title = "Simulation vs Reference")
    display(plot!(Xm[1,:], Xm[2,:], label = "actual"))

end
```




```
Test Summary:   | Pass  Total  Time
sim to real gap |    2      2  0.0s
Test.DefaultTestSet("sim to real gap", Any[], 2, false, false, true, 1.711227034108244e9, 1.71122703415168e9, false)
```

In order to account for this, we are going to use ILC to iteratively correct our control until we converge.

To encourage the trajectory of the bike to follow the reference, the objective value for this problem is the following: $$ J(X,U) = \sum_{i=1}^{N-1} \bigg[\frac{1}{2} (x_i - x_{ref,i})^TQ(x_i - x_{ref,i}) + \frac{1}{2} (u_i-u_{ref,i})^TR(u_i-u_{ref,i}) \bigg] + \frac{1}{2}(x_N - x_{ref,N})^TQ_f(x_N - x_{ref,N})$$

Using ILC as described in Lecture 18, we are to linearize our approximate dynamics model about X_{ref} and U_{ref} to get the following Jacobians:

$$ A_k = \frac{\partial f}{\partial x}\bigg|_{x_{ref,k},u_{ref,k}}, \quad \quad B_k = \frac{\partial f}{\partial u}\bigg|_{x_{ref,k},u_{ref,k}} $$

where $f(x,u)$ is our **approximate discrete** dynamics model (`estimated_car_dynamics` + `rk4`). **You will form these Jacobians exactly once, using `Xref` and `Uref` .** Here is a summary of the notation:

- X_{ref} (`Xref`) - Optimal trajectory computed offline with approximate dynamics model.
- U_{ref} (`Uref`) - Optimal controls computed offline with approximate dynamics model.
- X_{sim} (`Xsim`) - Simulated trajectory with real dynamics model.
- $\bar{U}$ (`Ubar`) - Control we use for simulation with real dynamics model (this is what ILC updates).

In the second step of ILC, we solve the following optimization problem:

$$ \begin{align} \min_{\Delta x_{1:N},\Delta u_{1:N-1}} \quad & J(X_{sim} + \Delta X, \bar{U} + \Delta U)\\ \text{st} \quad & \Delta x_1 = 0 \\ & \Delta x_{k+1} = A_k \Delta x_k + B_k \Delta u_k \quad \text{for } k = 1,2,\ldots,N-1 \\ \end{align}$$

We are going to initialize our $\bar{U}$ with U_{ref}, then the ILC algorithm will update $\bar{U} = \bar{U} + \Delta U$ at each iteration. It should only take 5-10 iterations to converge down to $\|\Delta U\| < 1 \cdot 10^{-2}$. You do not need to do any sort of linesearch between ILC updates.

In [ ]:
```
# feel free to use/not use any of these

function trajectory_cost(Xsim::Vector{Vector{Float64}}, # simulated states
                         Ubar::Vector{Vector{Float64}}, # simulated controls (ILC iterates this)
                         Xref::Vector{Vector{Float64}}, # reference X's we want to track
                         Uref::Vector{Vector{Float64}}, # reference U's we want to track
                         Q::Matrix,                     # LQR tracking cost term
                         R::Matrix,                     # LQR tracking cost term
                         Qf::Matrix                     # LQR tracking cost term
                         )::Float64                     # return cost J

    J = 0
    # TODO: return trajectory cost J(Xsim, Ubar)
    for i in 1:length(Ubar)
        J += (Xsim[i] - Xref[i])'*Q*(Xsim[i] - Xref[i]) + (Ubar[i] - Uref[i])'*R*(Ubar[i] - Uref[i])
    end
    J += (Xsim[end] - Xref[end])'*Qf*(Xsim[end] - Xref[end])
end

function vec_from_mat(Xm::Matrix)::Vector{Vector{Float64}}
    # convert a matrix into a vector of vectors
    X = [Xm[:,i] for i = 1:size(Xm,2)]
    return X
end

function ilc_update(Xsim::Vector{Vector{Float64}}, # simulated states
                    Ubar::Vector{Vector{Float64}}, # simulated controls (ILC iterates this)
                    Xref::Vector{Vector{Float64}}, # reference X's we want to track
                    Uref::Vector{Vector{Float64}}, # reference U's we want to track
                    As::Vector{Matrix{Float64}},   # vector of A jacobians at each time step
                    Bs::Vector{Matrix{Float64}},   # vector of B jacobians at each time step
                    Q::Matrix,                     # LQR tracking cost term
```

```julia
                    R::Matrix,                       # LQR tracking cost term
                    Qf::Matrix                       # LQR tracking cost term
                    )::Vector{Vector{Float64}}       # return vector of ΔU's

    # solve optimization problem for ILC update
    N = length(Xsim)
    nx,nu = size(Bs[1])

    # create variables
    ΔX = cvx.Variable(nx, N)
    ΔU = cvx.Variable(nu, N-1)

    # TODO: cost function (tracking cost on Xref, Uref)
    cost = 0
    for i in 1:N-1
        cost += cvx.square(cvx.norm(Q^0.5 * (ΔX[:,i] - (Xref[i] - Xsim[i])))) +
            cvx.square(cvx.norm(R^0.5 * (ΔU[:,i] - (Uref[i] - Ubar[i]))))
    end
    cost += cvx.square(cvx.norm(Qf^0.5 * (ΔX[:,N] - (Xref[N] - Xsim[N]))))

    # problem instance
    prob = cvx.minimize(cost)

    # TODO: initial condition constraint
    prob.constraints += (ΔX[:,1] == zeros(nx))

    # TODO: dynamics constraints
    for i in 1:N-1
        prob.constraints += (ΔX[:,i+1] == (As[i]*ΔX[:,i] + Bs[i]*ΔU[:,i]))
    end

    cvx.solve!(prob, ECOS.Optimizer; silent_solver = true)

    # return ΔU
    ΔU = vec_from_mat(ΔU.value)

    return ΔU
end
```

```
ilc_update (generic function with 1 method)
```

Here you will run your ILC algorithm. The resulting plots should show the simulated trajectory `Xsim` tracks `Xref` very closely, but there should be a significant difference between `Uref` and `Ubar`.

In [ ]:
```julia
@testset "ILC" begin

    # problem size
    nx = 5
    nu = 2
    dt = 0.1
    tf = 5.0
    t_vec = 0:dt:tf
    N = length(t_vec)

    # optimal trajectory computed offline with approximate model
    Xref, Uref = load_car_trajectory()

    # initial and terminal conditions
```

```julia
    xic = Xref[1]
    xg = Xref[N]

    # LQR tracking cost to be used in ILC
    Q = diagm([1,1,.1,.1,.1])
    R = .1*diagm(ones(nu))
    Qf = 1*diagm(ones(nx))

    # load all useful things into params
    model = (L = 2.8, lr = 1.6)

    params = (Q = Q, R = R, Qf = Qf,xic = xic, xg = xg, Xref=Xref,Uref=Uref,
          dt = dt,
          N = N,
          model = model)


    # this holds the sim trajectory (with real dynamics)
    Xsim = [zeros(nx) for i = 1:N]
    Xsim[1] = xic

    # this is the feedforward control ILC is updating
    Ubar = [zeros(nu) for i = 1:(N-1)]
    Ubar .= Uref # initialize Ubar with Uref

    # TODO: calculate Jacobians
    As = [zeros(nx,nx) for i = 1:N-1]
    Bs = [zeros(nx,nu) for i = 1:N-1]
    for i in 1:N-1
        x = Xref[i]
        u = Uref[i]
        A = FD.jacobian(_x -> rk4(model, true_car_dynamics, _x, u, dt), x)
        B = FD.jacobian(_u -> rk4(model, true_car_dynamics, x, _u, dt), u)
        As[i] = A
        Bs[i] = B
    end

    # logging stuff
    @printf "iter     objv        |ΔU|     \n"
    @printf "-----------------------------\n"

    for ilc_iter = 1:10 # it should not take more than 10 iterations to converge

        # TODO: rollout
        for i in 1:N-1
            Xsim[i+1] = rk4(model, true_car_dynamics, Xsim[i], Ubar[i], dt)
        end

        # TODO: calculate objective val (trajectory_cost)
        obj_val = trajectory_cost(Xsim, Ubar, Xref, Uref, Q, R, Qf)

        # solve optimization problem for update (ilc_update)
        ΔU = ilc_update(Xsim, Ubar, Xref, Uref, As, Bs, Q, R, Qf)

        # TODO: update the control
        Ubar = Ubar + ΔU

        # logging
```

```
        @printf("%3d   %10.3e  %10.3e  \n", ilc_iter, obj_val, sum(norm.(ΔU)))

    end

    # --------plotting/animation-------
    Xm= hcat(Xsim...)
    Um = hcat(Ubar...)
    Xrefm = hcat(Xref...)
    Urefm = hcat(Uref...)
    plot(Xrefm[1,:], Xrefm[2,:], ls = :dash, label = "reference",
         xlabel = "x (m)", ylabel = "y (m)", title = "Trajectory")
    display(plot!(Xm[1,:], Xm[2,:], label = "actual"))

    plot(t_vec[1:end-1], Urefm', ls = :dash, lc = [:green :blue],label = "",
         xlabel = "time (s)", ylabel = "controls", title = "Controls (-- is reference)")
    display(plot!(t_vec[1:end-1], Um', label = ["δ" "a"], lc = [:green :blue]))

    # animation
    vis = Visualizer()
    vis_traj!(vis, :traj, [[x[1],x[2],0.1] for x in Xsim]; R = 0.02)
    build_car!(vis[:car])
    anim = mc.Animation(floor(Int,1/dt))
    for k = 1:N
        mc.atframe(anim, k) do
            update_car_pose!(vis[:car], Xsim[k])
        end
    end
    mc.setanimation!(vis, anim)
    display(render(vis))

    # ----------testing-------------
    @test 0.1 <= sum(norm.(Xsim - Xref)) <= 1.0 # should be ~0.7
    @test 5 <=  sum(norm.(Ubar - Uref)) <= 10 # should be ~7.7

end
```

```
iter     objv        |ΔU|
------------------------------
  1    2.872e+03   6.701e+01
  2    1.794e+03   3.614e+01
  3    1.590e+03   4.016e+01
  4    9.646e+02   1.929e+01
  5    5.250e+02   3.530e+01
  6    1.471e+02   1.646e+01
  7    1.997e+01   9.419e+00
  8    5.618e-01   1.212e+00
  9    1.429e-01   2.535e-02
 10    1.428e-01   1.815e-04
Test Summary: | Pass  Total  Time
ILC           |    2      2  1.4s
```


Trajectory
reference
actual
y (m)
x (m)


Controls (-- is reference)
δ
a
controls
time (s)

```
Test.DefaultTestSet("ILC", Any[], 2, false, false, true, 1.711227655894654e9, 1.711227657302288e9, false)
```

In [ ]:
```
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import MathOptInterface as MOI
import Ipopt
import FiniteDiff
import ForwardDiff as FD
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
using MeshCat
const mc = MeshCat
using StaticArrays
using Printf
```

```
 Activating project at `~/Course/CMU16-745-Optimal-Control-HW/hw4`
```

Julia note:

incorrect:

`x_l[idx.x[i]][2] = 0 # this does not change x_l`

correct:

`x_l[idx.x[i][2]] = 0 # this changes x_l`

It should always be `v[index] = new_val` if I want to update v with `new_val` at `index`.

In [ ]:
```
let

    # vector we want to modify
    Z = randn(5)

    # original value of Z so we can check if we are changing it
    Z_original = 1 * Z

    # index range we are considering
    idx_x = 1:3

    # this does NOT change Z
    Z[idx_x][2] = 0

    # we can prove this
    @show norm(Z - Z_original)

    # this DOES change Z
    Z[idx_x[2]] = 0

    # we can prove this
```

```
    @show norm(Z - Z_original)


end
```

```
norm(Z - Z_original) = 0.0
norm(Z - Z_original) = 0.2633874015670601
0.2633874015670601
```

In [ ]:
```
include(joinpath(@__DIR__, "utils","fmincon.jl"))
include(joinpath(@__DIR__, "utils","walker.jl"))
```

```
update_walker_pose! (generic function with 1 method)
```

(If nothing loads here, check out `walker.gif` in the repo)

NOTE: This question will have long outputs for each cell, remember you can use `cell -> all output -> toggle scrolling` to better see it all

Q2: Hybrid Trajectory Optimization (60 pts)

In this problem you'll use a direct method to optimize a walking trajectory for a simple biped model, using the hybrid dynamics formulation. You'll pre-specify a gait sequence and solve the problem using Ipopt. Your final solution should look like the video above.

The Dynamics

Our system is modeled as three point masses: one for the body and one for each foot. The state is defined as the x and y positions and velocities of these masses, for a total of 6 degrees of freedom and 12 states. We will label the position and velocity of each body with the following notation: *[Math Processing Error]* Each leg is connected to the body with prismatic joints. The system has three control inputs: a force along each leg, and the torque between the legs.

The state and control vectors are ordered as follows:

$$x = \begin{matrix} p_x^{(b)} \\ p_y^{(b)} \\ p_x^{(1)} \\ p_y^{(1)} \\ p_x^{(2)} \\ p_y^{(2)} \\ v_x^{(b)} \\ v_y^{(b)} \\ v_x^{(1)} \\ v_y^{(1)} \\ v_x^{(2)} \\ v_y^{(2)} \end{matrix} \qquad u = \begin{matrix} F^{(1)} \\ F^{(2)} \\ \tau \end{matrix}$$

where e.g.
$p_x^{(b)}$
is the
x
position of the body,
$v_y^{(i)}$
is the
y
velocity of foot
i
,
$F^{(i)}$
is the force along leg
i
, and
τ
is the torque between the legs.

The continuous time dynamics and jump maps for the two stances are shown below:

In [ ]:
```
function stance1_dynamics(model::NamedTuple, x::Vector, u::Vector)
    # dynamics when foot 1 is in contact with the ground

    mb,mf = model.mb, model.mf
    g = model.g

    M = Diagonal([mb mb mf mf mf mf])

    rb  = x[1:2]   # position of the body
    rf1 = x[3:4]   # position of foot 1
    rf2 = x[5:6]   # position of foot 2
    v   = x[7:12]  # velocities


    ℓ1x = (rb[1]-rf1[1])/norm(rb-rf1)
    ℓ1y = (rb[2]-rf1[2])/norm(rb-rf1)
    ℓ2x = (rb[1]-rf2[1])/norm(rb-rf2)
    ℓ2y = (rb[2]-rf2[2])/norm(rb-rf2)

    B = [ℓ1x  ℓ2x  ℓ1y-ℓ2y;
         ℓ1y  ℓ2y  ℓ2x-ℓ1x;
          0    0     0;
          0    0     0;
          0  -ℓ2x  ℓ2y;
          0  -ℓ2y -ℓ2x]

    v̇ = [0; -g; 0; 0; 0; -g] + M\(B*u)

    ẋ = [v; v̇]

    return ẋ
end

function stance2_dynamics(model::NamedTuple, x::Vector, u::Vector)
    # dynamics when foot 2 is in contact with the ground
```

```julia
    mb,mf = model.mb, model.mf
    g = model.g
    M = Diagonal([mb mb mf mf mf mf])

    rb  = x[1:2]   # position of the body
    rf1 = x[3:4]   # position of foot 1
    rf2 = x[5:6]   # position of foot 2
    v   = x[7:12]  # velocities

    ℓ1x = (rb[1]-rf1[1])/norm(rb-rf1)
    ℓ1y = (rb[2]-rf1[2])/norm(rb-rf1)
    ℓ2x = (rb[1]-rf2[1])/norm(rb-rf2)
    ℓ2y = (rb[2]-rf2[2])/norm(rb-rf2)

    B = [ℓ1x  ℓ2x  ℓ1y-ℓ2y;
         ℓ1y  ℓ2y  ℓ2x-ℓ1x;
        -ℓ1x   0  -ℓ1y;
        -ℓ1y   0   ℓ1x;
          0    0    0;
          0    0    0]

    v̇ = [0; -g; 0; -g; 0; 0] + M\(B*u)

    ẋ = [v; v̇]

    return ẋ
end

function jump1_map(x)
    # foot 1 experiences inelastic collision
    xn = [x[1:8]; 0.0; 0.0; x[11:12]]
    return xn
end

function jump2_map(x)
    # foot 2 experiences inelastic collision
    xn = [x[1:10]; 0.0; 0.0]
    return xn
end

function rk4(model::NamedTuple, ode::Function, x::Vector, u::Vector, dt::Real)::Vector
    k1 = dt * ode(model, x,        u)
    k2 = dt * ode(model, x + k1/2, u)
    k3 = dt * ode(model, x + k2/2, u)
    k4 = dt * ode(model, x + k3,   u)
    return x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end
```

```
rk4 (generic function with 1 method)
```

We are setting up this problem by scheduling out the contact sequence. To do this, we will define the following sets:

[Math Processing Error]

where
$\mathcal{M}_1$
contains the time steps when foot 1 is pinned to the ground (`stance1_dynamics`), and

$\mathcal{M}_2$
contains the time steps when foot 2 is pinned to the ground (`stance2_dynamics`). The jump map sets
$\mathcal{J}_1$
and
$\mathcal{J}_2$
are the indices where the mode of the next time step is different than the current, i.e.
$\mathcal{J}_i \equiv \{k+1 \notin \mathcal{M}_i \mid$
. We can write these out explicitly as the following:

[Math Processing Error]

Another term you will see is set subtraction, or
$\mathcal{M}_i \setminus \mathcal{J}_i$
. This just means that if
$k \in \mathcal{M}_i \setminus \mathcal{J}_i$
, then
k
is in
$\mathcal{M}_i$
but not in
$\mathcal{J}_i$
.

We will make use of the following Julia code for determining which set an index belongs to:

In [ ]:
```
let
    M1 = vcat([ (i-1)*10      .+ (1:5)   for i = 1:5]...) # stack the set into a vector
    M2 = vcat([((i-1)*10 + 5) .+ (1:5)   for i = 1:4]...) # stack the set into a vector
    J1 = [5,15,25,35]
    J2 = [10,20,30,40]

    @show  (5 in M1) # show if 5 is in M1
    @show  (5 in J1) # show if 5 is in J1
    @show !(5 in M1) # show is 5 is not in M1

    @show (5 in M1) && !(5 in J1) # 5 in M1 but not J1 (5 ∈ M_1 \ J1)

end
```

```
5 in M1 = true
5 in J1 = true
!(5 in M1) = false
5 in M1 && !(5 in J1) = false
false
```

We are now going to setup and solve a constrained nonlinear program. The optimization problem looks complicated but each piece should make sense and be relatively straightforward to implement. First we have the following LQR cost function that will track
x_{ref}
(`Xref`) and
u_{ref}
(`Uref`):

$$J(x_{1:N}, u_{1:N-1}) = \sum_{i=1}^{N-1}$$

Which goes into the following full optimization problem: *[Math Processing Error]*

Each constraint is now described, with the type of constraint for `fmincon` in parantheses:

1. Initial condition constraint **(equality constraint)**.
2. Terminal condition constraint **(equality constraint)**.
3. Stance 1 discrete dynamics **(equality constraint)**.
4. Stance 2 discrete dynamics **(equality constraint)**.
5. Discrete dynamics from stance 1 to stance 2 with jump 2 map **(equality constraint)**.
6. Discrete dynamics from stance 2 to stance 1 with jump 1 map **(equality constraint)**.
7. Make sure the foot 1 is pinned to the ground in stance 1 **(equality constraint)**.
8. Make sure the foot 2 is pinned to the ground in stance 2 **(equality constraint)**.
9. Length constraints between main body and foot 1 **(inequality constraint)**.
10. Length constraints between main body and foot 2 **(inequality constraint)**.
11. Keep the y position of all 3 bodies above ground **(primal bound)**.

And here we have the list of mathematical functions to the Julia function names:

- f_1 is `stance1_dynamics` + `rk4`
- f_2 is `stance2_dynamics` + `rk4`
- g_1 is `jump1_map`
- g_2 is `jump2_map`

For instance, $g_2(f_1(x_k, u_k))$ is `jump2_map(rk4(model, stance1_dynamics, xk, uk, dt))`

Remember that $r^{(b)}$ is defined above.

In [ ]:

```julia
function reference_trajectory(model, xic, xg, dt, N)
    # creates a reference Xref and Uref for walker

    Uref = [[model.mb*model.g*0.5;model.mb*model.g*0.5;0] for i = 1:(N-1)]

    Xref = [zeros(12) for i = 1:N]
```

```
    horiz_v = (3/N)/dt
    xs = range(-1.5, 1.5, length = N)
    Xref[1] = 1*xic
    Xref[N] = 1*xg

    for i = 2:(N-1)
        Xref[i] = [xs[i],1,xs[i],0,xs[i],0,horiz_v,0,horiz_v,0,horiz_v,0]
    end

    return Xref, Uref
end
```

```
reference_trajectory (generic function with 1 method)
```

To solve this problem with Ipopt and `fmincon`, we are going to concatenate all of our x's and u's into one vector (same as HW3Q1):

$$Z = \begin{bmatrix} x_1 \\ u_1 \\ x_2 \\ u_2 \\ \vdots \\ x_{N-1} \\ u_{N-1} \\ x_N \end{bmatrix} \in \mathbb{R}^{N \cdot nx + (N-1) \cdot nu}$$

where $x \in \mathbb{R}^{nx}$ and $u \in \mathbb{R}^{nu}$. Below we will provide useful indexing guide in `create_idx` to help you deal with Z. Remember that the API for `fmincon` (that we used in HW3Q1) is the following: *[Math Processing Error]*

Template code has been given to solve this problem but you should feel free to do whatever is easiest for you, as long as you get the trajectory shown in the animation `walker.gif` and pass tests.

In [ ]:
```
# feel free to solve this problem however you like, below is a template for a
# good way to start.

function create_idx(nx,nu,N)
    # create idx for indexing convenience
    # x_i = Z[idx.x[i]]
    # u_i = Z[idx.u[i]]
    # and stacked dynamics constraints of size nx are
    # c[idx.c[i]] = <dynamics constraint at time step i>
    #
    # feel free to use/not use this

    # our Z vector is [x0, u0, x1, u1, …, xN]
```

```julia
    nz = (N-1) * nu + N * nx # length of Z
    x = [(i - 1) * (nx + nu) .+ (1 : nx) for i = 1:N]
    u = [(i - 1) * (nx + nu) .+ ((nx + 1):(nx + nu)) for i = 1:(N - 1)]

    # constraint indexing for the (N-1) dynamics constraints when stacked up
    c = [(i - 1) * (nx) .+ (1 : nx) for i = 1:(N - 1)]
    nc = (N - 1) * nx # (N-1)*nx

    return (nx=nx,nu=nu,N=N,nz=nz,nc=nc,x= x,u = u,c = c)
end

function walker_cost(params::NamedTuple, Z::Vector)::Real
    # cost function
    idx, N, xg = params.idx, params.N, params.xg
    Q, R, Qf = params.Q, params.R, params.Qf
    Xref,Uref = params.Xref, params.Uref

    # TODO: input walker LQR cost

    J = 0
    for i = 1:(N-1)
        x = Z[idx.x[i]]
        u = Z[idx.u[i]]
        J += 0.5*(x - Xref[i])'*Q*(x - Xref[i]) + 0.5*u'*R*u
    end
    xf = Z[idx.x[N]]
    J += 0.5*(xf - xg)'*Qf*(xf - xg)


    return J
end

function walker_dynamics_constraints(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt
    M1, M2 = params.M1, params.M2
    J1, J2 = params.J1, params.J2
    model = params.model

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), idx.nc)

    # TODO: input walker dynamics constraints (constraints 3-6 in the opti problem)
    for i = 1:(N-1)
        x = Z[idx.x[i]]
        x_next = Z[idx.x[i+1]]
        u = Z[idx.u[i]]
        if (i in M1) && !(i in J1)
            c[idx.c[i]] = (x_next - rk4(model,stance1_dynamics,x,u,dt))
        end
        if (i in M2) && !(i in J2)
            c[idx.c[i]] = (x_next - rk4(model,stance2_dynamics,x,u,dt))
        end
        if i in J1
            c[idx.c[i]] = (x_next - jump2_map(rk4(model,stance1_dynamics,x,u,dt)))
        end
        if i in J2
            c[idx.c[i]] = (x_next - jump1_map(rk4(model,stance2_dynamics,x,u,dt)))
        end
```

```julia
    end

    return c
end

function walker_stance_constraint(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt
    M1, M2 = params.M1, params.M2
    J1, J2 = params.J1, params.J2

    model = params.model

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), N)

    # TODO: add walker stance constraints (constraints 7-8 in the opti problem)
    for i = 1:N
        x = Z[idx.x[i]]
        if i in M1
            c[i] = x[4] # foot 1 in contact with the ground
        end
        if i in M2
            c[i] = x[6] # foot 2 in contact with the ground
        end
    end

    return c
end

function walker_equality_constraint(params::NamedTuple, Z::Vector)::Vector
    N, idx, xic, xg = params.N, params.idx, params.xic, params.xg

    # TODO: stack up all of our equality constraints

    # should be length 2*nx + (N-1)*nx + N
    # inital condition constraint (nx)       (constraint 1)
    c_init = Z[idx.x[1]] - xic

    # terminal constraint         (nx)       (constraint 2)
    c_term = Z[idx.x[N]] - xg

    # dynamics constraints        (N-1)*nx   (constraint 3-6)
    c_dyn = walker_dynamics_constraints(params, Z)

    # stance constraint           N          (constraint 7-8)
    c_stance = walker_stance_constraint(params, Z)

    return [c_init; c_term; c_dyn; c_stance]
end

function walker_inequality_constraint(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt
    M1, M2 = params.M1, params.M2

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), 2*N)

    # TODO: add the length constraints shown in constraints (9-10)
```

```
    # there are 2*N constraints here
    for i = 1:N
        rb = Z[idx.x[i]][1:2]
        r1 = Z[idx.x[i]][3:4]
        r2 = Z[idx.x[i]][5:6]
        d1 = norm(rb-r1)
        d2 = norm(rb-r2)
        c[2*i-1] = d1
        c[2*i] = d2
    end

    return c
end
```

walker_inequality_constraint (generic function with 1 method)

In [ ]:
```
@testset "walker trajectory optimization" begin

    # dynamics parameters
    model = (g = 9.81, mb= 5.0, mf = 1.0, ℓ_min = 0.5, ℓ_max = 1.5)

    # problem size
    nx = 12
    nu = 3
    tf = 4.4
    dt = 0.1
    t_vec = 0:dt:tf
    N = length(t_vec)

    # initial and goal states
    xic = [-1.5;1;-1.5;0;-1.5;0;0;0;0;0;0;0]
    xg =  [1.5;1;1.5;0;1.5;0;0;0;0;0;0;0]

    # index sets
    M1 = vcat([ (i-1)*10      .+ (1:5)   for i = 1:5]...)
    M2 = vcat([((i-1)*10 + 5) .+ (1:5)   for i = 1:4]...)
    J1 = [5,15,25,35]
    J2 = [10,20,30,40]

    # reference trajectory
    Xref, Uref = reference_trajectory(model, xic, xg, dt, N)

    # LQR cost function (tracking Xref, Uref)
    Q = diagm([1; 10; fill(1.0, 4); 1; 10; fill(1.0, 4)]);
    R = diagm(fill(1e-3,3))
    Qf = 1*Q;

    # create indexing utilities
    idx = create_idx(nx,nu,N)

    # put everything useful in params
    params = (
        model = model,
        nx = nx,
        nu = nu,
        tf = tf,
        dt = dt,
        t_vec = t_vec,
        N = N,
```

```julia
        M1 = M1,
        M2 = M2,
        J1 = J1,
        J2 = J2,
        xic = xic,
        xg = xg,
        idx = idx,
        Q = Q, R = R, Qf = Qf,
        Xref = Xref,
        Uref = Uref
    )

    # TODO: primal bounds (constraint 11)
    x_l = -Inf*ones(idx.nz) # update this
    x_u =  Inf*ones(idx.nz) # update this
    for i = 1:N
        x_l[idx.x[i][2]] = 0
        x_l[idx.x[i][4]] = 0
        x_l[idx.x[i][6]] = 0
    end

    # TODO: inequality constraint bounds
    c_l = 0.5*ones(2*N) # update this
    c_u = 1.5*ones(2*N) # update this

    # TODO: initialize z0 with the reference Xref, Uref
    z0 = zeros(idx.nz) # update this
    for i = 1:N
        z0[idx.x[i]] = Xref[i]
    end
    for i = 1:(N-1)
        z0[idx.u[i]] = Uref[i]
    end

    # adding a little noise to the initial guess is a good idea
    z0 = z0 + (1e-6)*randn(idx.nz)

    diff_type = :auto

    Z = fmincon(walker_cost,walker_equality_constraint,walker_inequality_constraint,
                x_l,x_u,c_l,c_u,z0,params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = true)

    # pull the X and U solutions out of Z
    X = [Z[idx.x[i]] for i = 1:N]
    U = [Z[idx.u[i]] for i = 1:(N-1)]

    # ------------plotting--------------
    Xm = hcat(X...)
    Um = hcat(U...)

    plot(Xm[1,:],Xm[2,:], label = "body")
    plot!(Xm[3,:],Xm[4,:], label = "leg 1")
    display(plot!(Xm[5,:],Xm[6,:], label = "leg 2",xlabel = "x (m)",
                  ylabel = "y (m)", title = "Body Positions"))

    display(plot(t_vec[1:end-1], Um',xlabel = "time (s)", ylabel = "U",
                 label = ["F1" "F2" "τ"], title = "Controls"))
```

```
    # ---------animation-------------
    vis = Visualizer()
    build_walker!(vis, model::NamedTuple)
    anim = mc.Animation(floor(Int,1/dt))
    for k = 1:N
        mc.atframe(anim, k) do
            update_walker_pose!(vis, model::NamedTuple, X[k])
        end
    end
    mc.setanimation!(vis, anim)
    display(render(vis))

    # --------testing---------------

    # initial and terminal states
    @test norm(X[1] - xic,Inf) <= 1e-3
    @test norm(X[end] - xg,Inf) <= 1e-3

    for x in X

        # distance between bodies
        rb = x[1:2]
        rf1 = x[3:4]
        rf2 = x[5:6]
        @test (0.5 - 1e-3) <= norm(rb-rf1) <= (1.5 + 1e-3)
        @test (0.5 - 1e-3) <= norm(rb-rf2) <= (1.5 + 1e-3)

        # no two feet moving at once
        v1 = x[9:10]
        v2 = x[11:12]
        @test min(norm(v1,Inf),norm(v2,Inf)) <= 1e-3

        # check everything above the surface
        @test x[2] >= (0 - 1e-3)
        @test x[4] >= (0 - 1e-3)
        @test x[6] >= (0 - 1e-3)

    end

end
```

```
---------checking dimensions of everything----------
---------all dimensions good-----------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient----------------
---------testing constraint Jacobian---------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve---------------------
This is Ipopt version 3.14.14, running with linear solver MUMPS 5.6.2.

Number of nonzeros in equality constraint Jacobian...:   401184
Number of nonzeros in inequality constraint Jacobian.:    60480
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:      672
                     variables with only lower bounds:      135
                variables with lower and upper bounds:        0
                     variables with only upper bounds:        0
```

```
Total number of equality constraints.................:      597
Total number of inequality constraints...............:       90
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:       90
        inequality constraints with only upper bounds:        0

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  2.6469427e+01 1.47e+00 1.00e+00   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  1.2108665e+02 1.05e+00 5.66e+03  -0.7 1.18e+02    -  3.33e-01 4.07e-01h  1
   2  2.4342060e+02 5.60e-01 5.11e+03   0.4 8.87e+01    -  1.00e+00 4.56e-01h  1
   3  2.9219552e+02 5.56e-01 3.92e+03   0.7 9.16e+01    -  1.00e+00 3.36e-01f  2
   4  3.7019423e+02 3.34e-01 3.22e+03   0.9 4.05e+01    -  1.00e+00 4.00e-01h  1
   5  4.0053188e+02 1.24e-01 3.28e+02  -5.1 2.72e+01    -  5.68e-01 1.00e+00h  1
   6  3.5496382e+02 1.46e-01 1.41e+02  -5.5 4.86e+01    -  4.24e-01 9.91e-01f  1
   7  3.3285706e+02 2.78e-02 3.61e+03  -1.9 2.37e+01    -  5.67e-01 1.00e+00f  1
   8  3.2540607e+02 7.66e-03 5.45e+03  -1.0 8.93e+00    -  3.31e-01 1.00e+00f  1
   9  3.1279339e+02 1.44e-02 2.94e+03  -1.6 1.37e+01    -  4.24e-01 8.53e-01f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  10  3.0298271e+02 2.97e-02 5.15e+01  -0.8 1.92e+01    -  1.00e+00 1.00e+00f  1
  11  2.9157041e+02 6.72e-02 4.73e+03  -0.2 4.12e+01    -  1.00e+00 3.26e-01f  2
  12  2.8427163e+02 2.76e-02 7.15e+00  -0.6 2.07e+01    -  1.00e+00 1.00e+00f  1
  13  2.8361011e+02 4.35e-03 5.64e+01  -1.2 2.19e+01    -  9.50e-01 1.00e+00H  1
  14  2.8293452e+02 2.49e-02 5.24e+00  -1.3 1.21e+01    -  1.00e+00 1.00e+00f  1
  15  2.7803134e+02 1.44e-02 5.97e+00  -1.6 1.01e+01    -  1.00e+00 1.00e+00f  1
  16  2.7547486e+02 3.53e-03 1.35e+00  -2.2 5.75e+00    -  1.00e+00 1.00e+00f  1
  17  2.7491497e+02 1.01e-03 1.05e+00  -3.2 2.71e+00    -  1.00e+00 1.00e+00f  1
  18  2.7396553e+02 2.78e-03 4.36e+00  -4.1 7.83e+00    -  1.00e+00 1.00e+00f  1
  19  2.7341890e+02 9.73e-03 1.06e+01  -3.8 8.17e+01    -  1.00e+00 1.56e-01f  2
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  20  2.7308849e+02 1.56e-02 1.36e+01  -4.2 3.83e+01    -  1.00e+00 2.99e-01f  1
  21  2.7253655e+02 9.77e-03 1.10e+01  -3.6 7.35e+00    -  1.00e+00 6.09e-01f  1
  22  2.7166896e+02 2.71e-03 3.98e+01  -3.3 6.67e+00    -  2.46e-01 1.00e+00f  1
  23  2.7142457e+02 2.08e-03 1.13e+00  -3.2 4.12e+00    -  1.00e+00 1.00e+00f  1
  24  2.7122478e+02 2.48e-03 1.18e+00  -4.4 3.51e+00    -  1.00e+00 1.00e+00f  1
  25  2.7111154e+02 1.89e-03 1.46e+01  -5.4 8.09e+00    -  1.00e+00 4.14e-01f  1
  26  2.7110749e+02 1.88e-03 4.34e+01  -6.0 7.16e+00    -  1.00e+00 6.66e-03h  1
  27  2.7092304e+02 1.46e-03 3.83e+01  -5.9 8.62e+00    -  1.00e+00 4.02e-01f  1
  28  2.7290836e+02 3.57e-04 7.52e+00  -5.2 1.32e+01    -  1.00e+00 9.37e-01H  1
  29  2.7284015e+02 3.51e-04 3.84e+01  -5.7 3.32e+00    -  1.00e+00 1.51e-02f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  30  2.7072819e+02 3.35e-03 5.38e-01  -6.4 5.78e+00    -  1.00e+00 1.00e+00f  1
  31  2.7075920e+02 2.19e-05 1.94e-01  -6.9 3.37e-01    -  1.00e+00 1.00e+00h  1
  32  2.7075329e+02 1.17e-05 1.74e-01  -8.1 4.67e-01    -  1.00e+00 1.00e+00h  1
  33  2.7074994e+02 1.57e-05 1.69e-01  -7.4 2.80e-01    -  1.00e+00 1.00e+00h  1
  34  2.7074412e+02 1.55e-05 1.94e-01  -7.7 5.85e-01    -  1.00e+00 9.99e-01h  1
  35  2.7074279e+02 2.49e-05 8.64e+01  -8.8 1.47e+00    -  1.00e+00 5.00e-01h  2
  36  2.7073860e+02 4.11e-05 1.34e-01  -9.3 6.33e-01    -  1.00e+00 1.00e+00h  1
  37  2.7074777e+02 1.07e-07 2.56e-01 -10.6 6.31e-01    -  1.00e+00 1.00e+00H  1
  38  2.7073687e+02 2.68e-05 6.48e-02  -9.8 2.90e-01    -  1.00e+00 1.00e+00f  1
  39  2.7073670e+02 4.55e-07 1.42e-02 -10.6 4.93e-02    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  40  2.7073675e+02 1.00e-08 2.75e-02 -11.0 1.57e-01    -  1.00e+00 1.00e+00H  1
  41  2.7073649e+02 1.10e-06 5.62e+02 -11.0 3.34e-01    -  1.00e+00 5.00e-01h  2
  42  2.7073779e+02 1.02e-08 2.03e-01 -11.0 2.49e-01    -  1.00e+00 1.00e+00H  1
  43  2.7073642e+02 5.15e-06 2.44e-02 -11.0 2.08e-01    -  1.00e+00 1.00e+00f  1
  44  2.7073682e+02 1.87e-06 3.95e-02 -11.0 6.07e-02    -  1.00e+00 1.00e+00h  1
  45  2.7073636e+02 1.17e-06 5.41e-03 -11.0 4.99e-02    -  1.00e+00 1.00e+00h  1
  46  2.7073635e+02 1.45e-08 4.16e-03 -11.0 7.29e-03    -  1.00e+00 1.00e+00h  1
```

```
  47  2.7073635e+02 1.00e-08 9.11e-04 -11.0 3.68e-03    -  1.00e+00 1.00e+00h  1
  48  2.7073635e+02 1.00e-08 1.81e-03 -11.0 1.16e-02    -  1.00e+00 1.00e+00h  1
  49  2.7073640e+02 1.00e-08 8.58e-03 -11.0 5.38e-02    -  1.00e+00 1.00e+00H  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  50  2.7073634e+02 1.53e-07 3.98e-03 -11.0 1.15e-02    -  1.00e+00 1.00e+00h  1
  51  2.7073634e+02 1.00e-08 1.56e-03 -11.0 8.03e-03    -  1.00e+00 1.00e+00h  1
  52  2.7073634e+02 1.00e-08 8.68e-04 -11.0 3.18e-03    -  1.00e+00 1.00e+00h  1
  53  2.7073634e+02 1.00e-08 1.62e-03 -11.0 1.12e-02    -  1.00e+00 1.00e+00H  1
  54  2.7073635e+02 1.00e-08 7.05e-03 -11.0 1.75e-02    -  1.00e+00 1.00e+00H  1
  55  2.7073634e+02 1.47e-08 2.48e-03 -11.0 1.54e-02    -  1.00e+00 1.00e+00h  1
  56  2.7073634e+02 1.00e-08 2.97e-03 -11.0 4.61e-03    -  1.00e+00 1.00e+00h  1
  57  2.7073633e+02 1.00e-08 1.71e+02  -9.0 3.11e-03    -  8.48e-01 1.00e+00h  1
  58  2.7073633e+02 1.00e-08 5.60e+02  -9.2 9.83e-04    -  8.07e-04 1.00e+00h  1
  59  2.7073633e+02 1.24e-08 8.69e+02  -9.2 5.56e-04    -  4.45e-02 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  60  2.7073633e+02 1.00e-08 3.91e-05  -9.2 1.29e-04    -  1.00e+00 1.00e+00H  1
  61  2.7073633e+02 1.00e-08 5.17e-04 -11.0 3.27e-03    -  1.00e+00 1.00e+00h  1
  62  2.7073634e+02 1.00e-08 1.49e-03 -11.0 4.40e-03    -  1.00e+00 1.00e+00H  1
  63  2.7073633e+02 1.00e-08 1.20e-04 -11.0 4.59e-03    -  1.00e+00 1.00e+00h  1
  64  2.7073633e+02 1.00e-08 9.57e-05 -11.0 1.80e-04    -  1.00e+00 1.00e+00h  1
  65  2.7073633e+02 1.00e-08 6.16e+02  -9.0 1.22e-04    -  3.44e-01 1.00e+00h  1
  66  2.7073633e+02 1.00e-08 1.69e+03  -9.5 1.49e-04    -  4.83e-02 1.00e+00H  1
  67  2.7073633e+02 1.00e-08 1.86e+03  -9.5 8.37e-04    -  1.21e-02 5.00e-01h  2
  68  2.7073633e+02 1.00e-08 1.59e+03  -9.5 1.17e-03    -  1.20e-01 1.22e-04h 14
  69  2.7073633e+02 1.00e-08 1.36e+03  -9.5 9.73e-04    -  1.47e-01 1.47e-01s 22
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  70  2.7073633e+02 1.09e-08 5.37e+02  -9.5 1.02e-03    -  6.04e-01 6.04e-01s 22
  71  2.7073633e+02 1.00e-08 2.64e-04  -9.5 1.13e-03    -  1.00e+00 1.00e+00s 22
  72  2.7073633e+02 1.00e-08 2.99e+02  -9.5 9.55e-04    -  7.72e-01 0.00e+00S 22
  73  2.7073633e+02 1.00e-08 1.25e+02  -9.5 2.67e-04    -  4.26e-01 1.00e+00h  1
  74  2.7073633e+02 1.00e-08 2.19e-05  -9.5 1.13e-04    -  1.00e+00 1.00e+00h  1
  75  2.7073633e+02 1.00e-08 2.52e+02  -9.5 8.46e-05    -  3.19e-01 1.00e+00h  1
  76  2.7073633e+02 1.00e-08 5.35e+02  -9.5 3.70e-05    -  1.47e-01 1.00e+00h  1
  77  2.7073633e+02 1.24e-08 9.12e+02  -9.5 2.31e-05    -  4.06e-03 1.00e+00h  1
  78  2.7073633e+02 1.00e-08 1.66e+03  -9.5 6.33e-05    -  5.96e-02 1.00e+00H  1
  79  2.7073633e+02 2.21e-08 1.13e+03  -9.5 1.94e-04    -  5.57e-01 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  80  2.7073633e+02 1.00e-08 1.59e+03  -9.5 1.01e-04    -  4.08e-03 1.00e+00h  1
  81  2.7073633e+02 1.00e-08 1.68e+03  -9.5 1.32e-04    -  1.51e-01 1.00e+00H  1
  82  2.7073633e+02 1.00e-08 2.35e+03  -9.5 4.62e-05    -  2.36e-02 1.00e+00H  1
  83  2.7073633e+02 9.99e-09 2.92e+03  -9.5 3.36e-05    -  3.84e-02 1.00e+00H  1
  84  2.7073633e+02 9.98e-09 3.27e+03  -9.5 5.33e-06    -  1.24e-01 1.00e+00H  1
  85  2.7073634e+02 1.14e-08 4.16e+03  -9.5 2.40e-06    -  1.71e-02 1.00e+00H  1
  86  2.7073634e+02 1.11e-08 3.13e+03  -9.5 3.11e-06    -  5.81e-02 1.00e+00H  1
  87  2.7073634e+02 2.86e-08 2.93e+03  -9.5 6.32e-05    -  7.76e-02 1.00e+00H  1
  88  2.7073633e+02 1.26e-07 4.26e+02  -9.5 1.26e-04    -  7.97e-01 1.00e+00h  1
  89  2.7073634e+02 5.82e-09 1.65e+00  -9.5 1.25e-04    -  9.97e-01 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  90  2.7073634e+02 5.82e-09 6.77e-06  -9.5 8.33e-06    -  1.00e+00 1.00e+00h  1
  91  2.7073634e+02 5.82e-09 1.82e-06 -11.0 5.32e-06    -  1.00e+00 1.00e+00h  1

Number of Iterations....: 91

                                   (scaled)                 (unscaled)
Objective...............:   2.7073633642841367e+02    2.7073633642841367e+02
Dual infeasibility......:   1.8156328731866456e-06    1.8156328731866456e-06
Constraint violation....:   5.8238018232080851e-09    5.8238018232080851e-09
Variable bound violation:   9.9999981001170043e-09    9.9999981001170043e-09
```

```
Complementarity.........:   1.5461256422574504e-11    1.5461256422574504e-11
Overall NLP error.......:   6.6002081195897654e-07    1.8156328731866456e-06


Number of objective function evaluations             = 175
Number of objective gradient evaluations             = 92
Number of equality constraint evaluations            = 175
Number of inequality constraint evaluations          = 175
Number of equality constraint Jacobian evaluations   = 92
Number of inequality constraint Jacobian evaluations = 92
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 30.627

EXIT: Optimal Solution Found.
```






```
┌ Info: Listening on: 127.0.0.1:8709, thread id: 1
└ @ HTTP.Servers /home/pcy/.julia/packages/HTTP/enKbm/src/Servers.jl:369
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
│ http://127.0.0.1:8709
└ @ MeshCat /home/pcy/.julia/packages/MeshCat/QXID5/src/visualizer.jl:64
```

```
Test Summary:                  | Pass  Total   Time
walker trajectory optimization |  272    272  32.0s
Test.DefaultTestSet("walker trajectory optimization", Any[], 272, false, false, true, 1.711230360784172e9, 1.711230392819971e9, false)
```

Q3 (5 pts)

Please fill out the following project form (one per group). This will primarily be for the TAs to use to understand what you are working on and hopefully be able to better assist you. If you haven't decided on certain aspects of the project, just include what you are currently thinking/what decisions you need to make.

(1) Write down your dynamics (handwritten, code, or latex). This can be continuous-time (include how you are discretizing your system) or discrete-time.

Our project study the dynamics of one drone with a slung attached to the bottom of it.

The drone dynamics is the same as the one we have studied in class, the only difference is that we have additional forces (which is extra control input) generated by the slung.

$$\dot{x} = \begin{bmatrix} \dot{r} \\ \dot{q} \\ \dot{v} \\ \dot{\omega} \end{bmatrix} = \begin{bmatrix} v \\ \frac{1}{2} q \otimes \hat{\omega} \\ g + \frac{1}{m^i}\left(R(q)F(u) + F_c\left(u_5, x, x^\ell\right)\right) \\ J^{-1}(\tau(u) - \omega \times J\omega) \end{bmatrix}$$

The load is modeled as a point mass:

$$\dot{x}^\ell = \begin{bmatrix} \dot{r}^\ell \\ \dot{v}^\ell \end{bmatrix} = \begin{bmatrix} v^\ell \\ g - \frac{1}{m^\ell} F_c(u_1, x, x^\ell)) \end{bmatrix}$$

Here the force is given by the following equation:

$$F_c\left(\gamma, x, x^\ell\right) = \gamma \frac{r^\ell - r}{\left\|r^\ell - r\right\|_2}$$

Also, to make sure the rope is always taut, we have the following constraint:

$$\left\| r - r^{\ell} \right\|_2 = d_{\text{cable}}$$

(2) What is your state (what does each variable represent)?

The state is the joint state of drone and the load. The state is defined as follows:

$$\bar{x} \in \mathbb{R}^{13+6} = \begin{bmatrix} x \\ x^{\ell} \end{bmatrix}$$

(3) What is your control (what does each variable represent)?

The control is the joint control of the drone and the load. The control is defined as follows:

$$\bar{u} \in \mathbb{R}^{5} = \begin{bmatrix} u \\ u^{\ell} \end{bmatrix}$$

(4) Briefly describe your goal for the project. What are you trying to make the system do? Specify whether you are doing control, trajectory optimization, both, or something else.

We are trying to do agile flight with a slung load attached to the drone. Specifically, we are trying to make the drone and the load passing a few gates in a certain order with trajectory optimization.

(5) What are your costs?

The cost is the quadratic cost to reach the desired position.

$$\tilde{x} = \bar{x} - x_{\text{des}}$$
$$J = \int_0^T \left(\tilde{x}^T Q \tilde{x} + \bar{u}^T R \bar{u}\right) dt$$

(6) What are your constraints?

The constrains has two parts: one for dynamics and another for collision avoidance.

For dynamics, we have the following constraints:

$$\begin{aligned} \text{taut cable} \quad & \left\| r - r^{\ell} \right\|_2 = d_{\text{cable}} \\ \text{cable control} \quad & (u)_5 = \left(u^{\ell}\right)_1 \\ \text{cable force} \quad & u^{\ell} \geq 0 \\ \text{control limits} \quad & 0 \leq (u)_j \leq u_{\max}, \quad \forall j \in \{1, \ldots, 4\} \end{aligned}$$

For collision avoidance, we have the following constraints:

$$\begin{aligned} \text{drone collision avoidance} \quad & d_{\text{quad}} + d_{\text{obs}} - \left\| p - p^j_{\text{obs}} \right\|_2 \leq 0 \\ \text{load collision avoidance} \quad & d_{\text{quad}} + d_{\text{obs}} - \left\| p_{\ell} - p^j_{\text{obs}} \right\|_2 \leq 0 \end{aligned}$$

(7) What solution methods are you going to try?

I am going to use direct collocation to solve the problem. The solver I am using is IPOPT.

(8) What have you tried so far?

I have tried use IPOPT to solve it and it now can navigate the drone to the desired position passing single door.

single door

(9) If applicable, what are you currently running into issues with?

How to scale the problem up to solve a multigate passing problem. For passing single door, the decision variable number is around 1k. For multiple doors, it would take long time to solve the problem.

(10) If your system doesn't fit with some of the questions above or there are additional things you'd like to elaborate on, please explain/do that here.