In [ ]:
```
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import MathOptInterface as MOI
import Ipopt
import FiniteDiff
import ForwardDiff
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
import MeshCat as mc
```

```
Activating project at `~/Desktop/2024Spring/CMU16745_OptimalControl/CMU16-745-Optimal-Control-HW/hw3`
```

In [ ]:
```
include(joinpath(@__DIR__, "utils","fmincon.jl"))
include(joinpath(@__DIR__, "utils","cartpole_animation.jl"))
```

Out[ ]:
```
animate_cartpole (generic function with 1 method)
```

NOTE: This question will have long outputs for each cell, remember you can use `cell -> all output -> toggle scrolling` to better see it all

Q1: Direct Collocation (DIRCOL) for a Cart Pole (30 pts)

We are now going to start working with the NonLinear Program (NLP) Solver IPOPT to solve some trajectory optimization problems. First we will demonstrate how this works for simple optimization problems (not trajectory optimization). The interface that we have setup for IPOPT is the following:

$$
\begin{aligned}
\min_x \quad & \ell(x) && \text{cost function} && (1)\\
\text{st} \quad & c_{eq}(x) = 0 && \text{equality constraint} && (2)\\
& c_L \leq c_{ineq}(x) \leq c_U && \text{inequality constraint} && (3)\\
& x_L \leq x \leq x_U && \text{primal bound constraint} && (4)
\end{aligned}
$$

where $\ell(x)$ is our objective function, $c_{eq}(x) = 0$ is our equality constraint, $c_L \leq c_{ineq}(x) \leq c_U$ is our bound inequality constraint, and $x_L \leq x \leq x_U$ is a bound constraint on our primal variable x.

Part A: Solve an LP with IPOPT (5 pts)

To demonstrate this, we are going to ask you to solve a simple Linear Program (LP):

$$
\begin{aligned}
\min_x \quad & q^T x && (5)\\
\text{st} \quad & Ax = b && (6)\\
& Gx \leq h && (7)
\end{aligned}
$$

Your job will be to transform this problem into the form shown above and solve it with IPOPT. To help you interface with IPOPT, we have created a function `fmincon` for you. Below is the docstring for this function that details all of the inputs.

In [ ]:

```
"""
x = fmincon(cost,equality_constraint,inequality_constraint,x_l,x_u,c_l,c_u,x0,params,diff_type)

This function uses IPOPT to minimize an objective function

`cost(params, x)`

With the following three constraints:

`equality_constraint(params, x) = 0`
`c_l <= inequality_constraint(params, x) <= c_u`
`x_l <= x <= x_u`

Note that the constraint functions should return vectors.
```

```
Problem specific parameters should be loaded into params::NamedTuple (things like
cost weights, dynamics parameters, etc.).

args:
    cost::Function                        - objective function to be minimzed (returns scalar)
    equality_constraint::Function         - c_eq(params, x) == 0
    inequality_constraint::Function       - c_l <= c_ineq(params, x) <= c_u
    x_l::Vector                           - x_l <= x <= x_u
    x_u::Vector                           - x_l <= x <= x_u
    c_l::Vector                           - c_l <= c_ineq(params, x) <= x_u
    c_u::Vector                           - c_l <= c_ineq(params, x) <= x_u
    x0::Vector                            - initial guess
    params::NamedTuple                    - problem parameters for use in costs/constraints
    diff_type::Symbol                     - :auto for ForwardDiff, :finite for FiniteDiff
    verbose::Bool                         - true for IPOPT output, false for nothing

optional args:
    tol                                   - optimality tolerance
    c_tol                                 - constraint violation tolerance
    max_iters                             - max iterations
    verbose                               - verbosity of IPOPT

outputs:
    x::Vector                             - solution

You should try and use :auto for your `diff_type` first, and only use :finite if you
absolutely cannot get ForwardDiff to work.

This function will run a few basic checks before sending the problem off to IPOPT to
solve. The outputs of these checks will be reported as the following:

---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------

If you're getting stuck during the testing of one of the derivatives, try switching
```

```
to FiniteDiff.jl by setting diff_type = :finite.
""";
```

In [ ]:
```julia
@testset "solve LP with IPOPT" begin

    LP = jldopen(joinpath(@__DIR__,"utils","random_LP.jld2"))

    params = (q = LP["q"], A = LP["A"], b = LP["b"], G = LP["G"], h = LP["h"])

    # return a scalar
    function cost(params, x)::Real
        # TODO: create cost function with params and x
        cost = params.q' * x
        return cost
    end

    # return a vector
    function equality_constraint(params, x)::Vector
        # TODO: create equality constraint function with params and x
        constrain = params.A * x - params.b
        return constrain
    end

    # return a vector
    function inequality_constraint(params, x)::Vector
        # TODO: create inequality constraint function with params and x
        constrain = params.G * x - params.h
        return constrain
    end

    # TODO: primal bounds
    # you may use Inf, like Inf*ones(10) for a vector of positive infinity
    x_l = -Inf * ones(20)
    x_u = Inf * ones(20)

    # TODO: inequality constraint bounds
    c_l = -Inf * ones(20)
    c_u = zeros(20)

    # initial guess
```

```julia
    x0= randn(20)

    diff_type = :auto   # use ForwardDiff.jl
#     diff_type = :finite # use FiniteDiff.jl

    x = fmincon(cost, equality_constraint, inequality_constraint,
                x_l, x_u, c_l, c_u, x0, params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = true);


    @test isapprox(x, [-0.44289, 0, 0, 0.19214, 0, 0, -0.109095,
                       -0.43221, 0, 0, 0.44289, 0, 0, 0.192142,
                       0, 0, 0.10909, 0.432219, 0, 0], atol = 1e-3)

end
```

```
---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------

******************************************************************************
This program contains Ipopt, a library for large-scale nonlinear optimization.
 Ipopt is released as open source code under the Eclipse Public License (EPL).
         For more information visit https://github.com/coin-or/Ipopt
******************************************************************************

This is Ipopt version 3.14.14, running with linear solver MUMPS 5.6.2.

Number of nonzeros in equality constraint Jacobian...:       80
Number of nonzeros in inequality constraint Jacobian.:      400
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:       20
                     variables with only lower bounds:        0
                variables with lower and upper bounds:        0
                     variables with only upper bounds:        0
Total number of equality constraints.................:        4
```

```
Total number of inequality constraints...............:       20
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:       20

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  6.1368111e+00 3.32e+00 3.33e-01   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  6.4609256e+00 3.73e-01 1.18e+00  -1.0 1.35e+00    -  6.38e-01 8.88e-01h  1
   2  2.1880591e+00 5.58e-02 1.77e-01  -6.5 2.20e+00    -  1.00e+00 8.51e-01f  1
   3  1.4894613e+00 2.06e-03 3.78e-02  -1.8 5.84e-01    -  7.78e-01 9.63e-01f  1
   4  1.2441002e+00 4.49e-04 4.49e-02  -3.2 1.16e-01    -  1.00e+00 7.82e-01f  1
   5  1.1878182e+00 5.55e-17 1.04e-02  -3.3 3.23e-02    -  9.60e-01 1.00e+00f  1
   6  1.1770767e+00 5.55e-17 1.62e-03  -5.1 4.87e-03    -  1.00e+00 9.30e-01f  1
   7  1.1763574e+00 9.71e-17 3.69e-12  -6.3 1.79e-04    -  1.00e+00 9.99e-01f  1

Number of Iterations....: 7

                                   (scaled)                 (unscaled)
Objective...............:   1.1763573802974470e+00    1.1763573802974470e+00
Dual infeasibility......:   3.6882473971854972e-12    3.6882473971854972e-12
Constraint violation....:   9.7144514654701197e-17    9.7144514654701197e-17
Variable bound violation:   0.0000000000000000e+00    0.0000000000000000e+00
Complementarity.........:   8.3998953991515956e-07    8.3998953991515956e-07
Overall NLP error.......:   8.3998953991515956e-07    8.3998953991515956e-07


Number of objective function evaluations             = 8
Number of objective gradient evaluations             = 8
Number of equality constraint evaluations            = 8
Number of inequality constraint evaluations          = 8
Number of equality constraint Jacobian evaluations   = 8
Number of inequality constraint Jacobian evaluations = 8
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 0.839

EXIT: Optimal Solution Found.
Test Summary:       | Pass  Total  Time
solve LP with IPOPT |    1      1  4.3s
```

Out[ ]:
```
Test.DefaultTestSet("solve LP with IPOPT", Any[], 1, false, false, true, 1.709237149964723e9, 1.709237154271613e9, false)
```

Part B: Cart Pole Swingup (20 pts)

We are now going to solve for a cartpole swingup. The state for the cartpole is the following:

$$x = [p, \theta, \dot{p}, \dot{\theta}]^T$$

Where p and θ can be seen in the graphic `cartpole.png`.



where we start with the pole in the down position ($\theta = 0$), and we want to use the horizontal force on the cart to drive the pole to the up position ($\theta = \pi$).

$$\begin{aligned}
\min_{x_{1:N}, u_{1:N-1}} \quad & \sum_{i=1}^{N-1} \left[\frac{1}{2}(x_i - x_{goal})^T Q (x_i - x_{goal}) + \frac{1}{2} u_i^T R u_i \right] + \frac{1}{2}(x_N - x_{goal})^T Q_f (x_N - x_{goal}) && (8) \\
\text{st} \quad & x_1 = x_{\text{IC}} && (9) \\
& x_N = x_{goal} && (10) \\
& f_{hs}(x_i, x_{i+1}, u_i, dt) = 0 \quad \text{for } i = 1, 2, \ldots, N-1 && (11) \\
& -10 \le u_i \le 10 \quad \text{for } i = 1, 2, \ldots, N-1 && (12)
\end{aligned}$$

Where $x_{IC} = [0, 0, 0, 0]$, and $x_{goal} = [0, \pi, 0, 0]$, and $f_{hs}(x_i, x_{i+1}, u_i)$ is the implicit integrator residual for Hermite Simpson (see HW1Q1 to refresh on this). Note that while Zac used a first order hold (FOH) on the controls in class (meaning we linearly interpolate controls between time steps), we are using a zero-order hold (ZOH) in this assignment. This means that each control u_i is held constant for the entirety of the timestep.

In [ ]:

```julia
# cartpole
function dynamics(params::NamedTuple, x::Vector, u)
    # cartpole ODE, parametrized by params.

    # cartpole physical parameters
    mc, mp, l = params.mc, params.mp, params.l
    g = 9.81

    q = x[1:2]
    qd = x[3:4]

    s = sin(q[2])
    c = cos(q[2])

    H = [mc+mp mp*l*c; mp*l*c mp*l^2]
    C = [0 -mp*qd[2]*l*s; 0 0]
    G = [0, mp*g*l*s]
    B = [1, 0]

    qdd = -H\(C*qd + G - B*u[1])
    xdot = [qd;qdd]
    return xdot

end
function hermite_simpson(params::NamedTuple, x1::Vector, x2::Vector, u, dt::Real)::Vector
    # TODO: input hermite simpson implicit integrator residual
    x_mid =
        0.5 * (x1 + x2) +
        0.125 * dt * (dynamics(params, x1, u) - dynamics(params, x2, u))
    return x1 +
        1 / 6 *
        dt *
        (
            dynamics(params, x1, u) +
```

```
                4 * dynamics(params, x_mid, u) +
                dynamics(params, x2, u)
            ) - x2
end
```

Out[ ]: hermite_simpson (generic function with 1 method)

To solve this problem with IPOPT and `fmincon`, we are going to concatenate all of our x's and u's into one vector:

$$Z = \begin{bmatrix} x_1 \\ u_1 \\ x_2 \\ u_2 \\ \vdots \\ x_{N-1} \\ u_{N-1} \\ x_N \end{bmatrix} \in \mathbb{R}^{N \cdot nx + (N-1) \cdot nu}$$

where $x \in \mathbb{R}^{nx}$ and $u \in \mathbb{R}^{nu}$. Below we will provide useful indexing guide in `create_idx` to help you deal with Z.

It is also worth noting that while there are inequality constraints present ($-10 \leq u_i \leq 10$), we do not need a specific `inequality_constraints` function as an input to `fmincon` since these are just bounds on the primal (Z) variable. You should use primal bounds in `fmincon` to capture these constraints.

In [ ]:
```
function create_idx(nx,nu,N)
    # This function creates some useful indexing tools for Z
    # x_i = Z[idx.x[i]]
    # u_i = Z[idx.u[i]]

    # Feel free to use/not use anything here.


    # our Z vector is [x0, u0, x1, u1, …, xN]
    nz = (N-1) * nu + N * nx # length of Z
    x = [(i - 1) * (nx + nu) .+ (1 : nx) for i = 1:N]
    u = [(i - 1) * (nx + nu) .+ ((nx + 1):(nx + nu)) for i = 1:(N - 1)]
```

```julia
    # constraint indexing for the (N-1) dynamics constraints when stacked up
    c = [(i - 1) * (nx) .+ (1 : nx) for i = 1:(N - 1)]
    nc = (N - 1) * nx # (N-1)*nx

    return (nx=nx,nu=nu,N=N,nz=nz,nc=nc,x= x,u = u,c = c)
end

function cartpole_cost(params::NamedTuple, Z::Vector)::Real
    idx, N, xg = params.idx, params.N, params.xg
    Q, R, Qf = params.Q, params.R, params.Qf

    # TODO: input cartpole LQR cost

    J = 0
    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]

        J += 0.5 * (xi - xg)' * Q * (xi - xg) + 0.5 * ui' * R * ui

    end

    # dont forget terminal cost
    J += 0.5 * (Z[idx.x[N]] - xg)' * Qf * (Z[idx.x[N]] - xg)

    return J
end

function cartpole_dynamics_constraints(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt

    # TODO: create dynamics constraints using hermite simpson

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), idx.nc)

    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]
```

```julia
        xip1 = Z[idx.x[i+1]]

        # TODO: hermite simpson
        c[idx.c[i]] = hermite_simpson(params, xi, xip1, ui, dt)
    end
    return c
end

function cartpole_equality_constraint(params::NamedTuple, Z::Vector)::Vector
    N, idx, xic, xg = params.N, params.idx, params.xic, params.xg

    # TODO: return all of the equality constraints
    c = zeros(eltype(Z), idx.nc + 8)

    # initial state constraint
    c[1:4] = Z[idx.x[1]] - xic

    # final state constraint
    c[5:8] = Z[idx.x[N]] - xg

    # dynamics constraints
    c[9:end] = cartpole_dynamics_constraints(params, Z)

    return c
end

function solve_cartpole_swingup(;verbose=true)

    # problem size
    nx = 4
    nu = 1
    dt = 0.05
    tf = 2.0
    t_vec = 0:dt:tf
    N = length(t_vec)

    # LQR cost
    Q = diagm(ones(nx))
    R = 0.1*diagm(ones(nu))
    Qf = 10*diagm(ones(nx))
```

```julia
    # indexing
    idx = create_idx(nx,nu,N)

    # initial and goal states
    xic = [0, 0, 0, 0]
    xg = [0, pi, 0, 0]

    # load all useful things into params
    params = (Q = Q, R = R, Qf = Qf, xic = xic, xg = xg, dt = dt, N = N, idx = idx,mc = 1.0, mp = 0.2, l

    # TODO: primal bounds
    x_l = zeros(idx.nz)
    x_u = zeros(idx.nz)
    for i = 1:(N-1)
        x_u[idx.u[i]] .= 10
        x_l[idx.u[i]] .= -10
    end
    for i = 1:N
        x_u[idx.x[i]] .= Inf
        x_l[idx.x[i]] .= -Inf
    end



    # inequality constraint bounds (this is what we do when we have no inequality constraints)
    c_l = zeros(0)
    c_u = zeros(0)
    function inequality_constraint(params, Z)
        return zeros(eltype(Z), 0)
    end

    # initial guess
    z0 = 0.001*randn(idx.nz)

    # choose diff type (try :auto, then use :finite if :auto doesn't work)
    diff_type = :auto
#     diff_type = :finite
```

```julia
    Z = fmincon(cartpole_cost,cartpole_equality_constraint,inequality_constraint,
                x_l,x_u,c_l,c_u,z0,params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = verbose)

    # pull the X and U solutions out of Z
    X = [Z[idx.x[i]] for i = 1:N]
    U = [Z[idx.u[i]] for i = 1:(N-1)]

    return X, U, t_vec, params
end

@testset "cartpole swingup" begin

    X, U, t_vec = solve_cartpole_swingup(verbose=true)


    # --------------testing------------------
    @test isapprox(X[1],zeros(4), atol = 1e-4)
    @test isapprox(X[end], [0,pi,0,0], atol = 1e-4)
    Xm = hcat(X...)
    Um = hcat(U...)

    # --------------plotting-----------------
    display(plot(t_vec, Xm', label = ["p" "θ" "ṗ" "θ̇"], xlabel = "time (s)", title = "State Trajectory"))
    display(plot(t_vec[1:end-1],Um',label="",xlabel = "time (s)", ylabel = "u",title = "Controls"))

    # meshcat animation
    display(animate_cartpole(X, 0.05))

end
```

```
---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------
This is Ipopt version 3.14.14, running with linear solver MUMPS 5.6.2.

Number of nonzeros in equality constraint Jacobian...:    34272
```

```
Number of nonzeros in inequality constraint Jacobian.:        0
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:      204
                     variables with only lower bounds:        0
                variables with lower and upper bounds:       40
                     variables with only upper bounds:        0
Total number of equality constraints.................:      168
Total number of inequality constraints...............:        0
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:        0

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  2.4671138e+02 3.14e+00 3.85e-04   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  2.7499972e+02 2.38e+00 7.99e+00  -5.0 1.28e+01    -  4.90e-01 2.43e-01h  3
   2  2.9805583e+02 2.16e+00 1.03e+01  -0.5 1.05e+01    -  6.11e-01 9.26e-02h  4
   3  3.3423733e+02 1.87e+00 1.40e+01  -0.4 1.29e+01    -  6.48e-01 1.33e-01h  3
   4  3.7117277e+02 1.61e+00 2.08e+01  -0.5 1.19e+01    -  8.80e-01 1.40e-01h  3
   5  4.1960392e+02 1.33e+00 2.73e+01  -0.8 1.00e+01    -  1.00e+00 1.74e-01h  3
   6  4.4376452e+02 1.20e+00 3.19e+01   0.3 1.84e+01    -  6.35e-01 9.61e-02h  3
   7  4.7560805e+02 1.07e+00 3.53e+01   0.2 1.80e+01    -  6.50e-01 1.12e-01h  3
   8  5.1181604e+02 9.43e-01 3.90e+01   0.3 2.25e+01    -  6.10e-01 1.17e-01h  3
   9  5.2145561e+02 8.53e-01 3.84e+01   0.3 1.15e+01    -  8.75e-01 9.51e-02h  3
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  10  5.1546760e+02 7.70e-01 4.12e+01   0.4 2.61e+01    -  5.18e-01 9.70e-02f  3
  11  5.0929550e+02 7.01e-01 4.40e+01   0.5 2.69e+01    -  6.05e-01 9.05e-02f  3
  12  5.0666365e+02 6.63e-01 4.55e+01   0.4 3.52e+01    -  8.46e-01 5.37e-02f  4
  13  5.1060601e+02 5.81e-01 5.23e+01   0.7 2.50e+01    -  3.09e-01 1.23e-01h  3
  14  5.3155453e+02 3.65e-01 8.03e+01  -5.3 1.97e+01    -  1.94e-01 3.71e-01H  1
  15  5.3201380e+02 3.38e-01 7.74e+01  -5.5 1.61e+01    -  2.75e-01 7.54e-02h  1
  16  5.4040117e+02 2.34e-01 7.03e+01   0.8 1.75e+01    -  4.44e-01 3.08e-01h  1
  17  5.4337905e+02 1.82e-01 7.45e+01   0.6 1.30e+01    -  5.84e-01 2.22e-01h  1
  18  5.4190602e+02 1.07e-01 8.48e+01   0.6 1.17e+01    -  7.77e-01 4.12e-01h  1
  19  5.4039350e+02 8.55e-02 8.08e+01   0.5 8.57e+00    -  9.86e-01 5.25e-01h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  20  5.2817601e+02 8.42e-02 4.62e+01   0.2 5.92e+00    -  9.59e-01 1.00e+00f  1
  21  5.0154129e+02 3.86e-02 1.91e+01   0.0 1.82e+00    -  9.84e-01 1.00e+00f  1
  22  4.8471830e+02 5.26e-02 2.23e+01  -0.0 1.21e+01    -  4.53e-01 2.17e-01f  1
  23  4.7170485e+02 4.39e-02 1.99e+01  -0.0 5.86e+00    -  8.85e-01 4.19e-01f  1
```

```
  24  4.6427517e+02 2.08e-01 4.18e+01  -0.3 3.08e+01    -  2.08e-01 2.63e-01f  1
  25  4.5200188e+02 5.96e-02 4.31e+01  -0.1 5.99e+00    -  8.83e-01 7.61e-01f  1
  26  4.4461906e+02 2.42e-03 2.18e+01  -0.5 2.12e+00    -  9.99e-01 1.00e+00f  1
  27  4.4208001e+02 1.42e-03 1.76e+01  -1.1 1.46e+00    -  1.00e+00 1.00e+00f  1
  28  4.4030196e+02 2.84e-03 1.85e+01  -1.5 1.59e+00    -  1.00e+00 1.00e+00f  1
  29  4.3824082e+02 9.18e-03 2.29e+01  -1.8 3.77e+00    -  1.00e+00 9.36e-01f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  30  4.3587661e+02 2.24e-02 1.91e+01  -0.9 9.20e+00    -  6.05e-01 1.00e+00F  1
  31  4.3429712e+02 3.31e-02 3.08e+01  -0.2 6.76e+00    -  1.00e+00 5.46e-01f  1
  32  4.3181248e+02 4.67e-03 1.54e+01  -1.1 1.96e+00    -  9.98e-01 9.30e-01f  1
  33  4.5423546e+02 1.37e-01 6.34e+00   0.9 2.44e+02    -  2.47e-01 5.96e-02f  1
  34  4.6099537e+02 3.72e-02 1.17e+01   0.3 8.69e+00    -  1.00e+00 9.38e-01f  1
  35  4.4307055e+02 5.70e-02 1.29e+01  -0.4 5.21e+00    -  7.55e-01 1.00e+00f  1
  36  4.3364936e+02 5.71e-02 1.46e+01  -0.4 4.89e+00    -  1.00e+00 1.00e+00f  1
  37  4.3295786e+02 5.79e-02 1.94e+01  -0.4 1.80e+01    -  1.00e+00 1.06e-01f  3
  38  4.2752238e+02 4.06e-03 1.40e+01  -0.4 1.58e+00    -  9.99e-01 1.00e+00f  1
  39  4.2505563e+02 6.45e-03 1.60e+01  -1.2 1.36e+00    -  1.00e+00 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  40  4.2373696e+02 4.35e-04 1.80e+01  -1.7 5.88e-01    -  9.98e-01 1.00e+00f  1
  41  4.2344595e+02 3.55e-04 1.84e+01  -2.9 3.30e-01    -  1.00e+00 1.00e+00f  1
  42  4.2334948e+02 1.13e-04 1.88e+01  -4.3 1.92e-01    -  1.00e+00 9.25e-01f  1
  43  4.2302630e+02 1.68e-02 2.36e+01  -1.9 8.71e+00    -  3.16e-01 3.86e-01f  1
  44  4.2244319e+02 5.24e-02 3.28e+01  -1.6 2.37e+01    -  9.89e-01 3.81e-01F  1
  45  4.1908908e+02 1.79e-03 2.26e+01  -1.5 3.64e+00    -  1.00e+00 1.00e+00f  1
  46  4.1889865e+02 2.14e-02 2.28e+01  -1.6 2.19e+00    -  1.00e+00 1.00e+00f  1
  47  4.1704693e+02 1.98e-02 2.26e+01  -1.4 5.44e+00    -  1.00e+00 3.12e-01f  1
  48  4.1898352e+02 7.19e-02 1.65e+01  -0.7 8.25e+00    -  1.00e+00 8.99e-01f  1
  49  4.1625873e+02 3.08e-02 2.62e+01  -0.8 4.16e+00    -  1.00e+00 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  50  4.1368758e+02 7.46e-03 2.47e+01  -0.5 2.58e+00    -  9.49e-01 1.00e+00f  1
  51  4.1037345e+02 1.53e-02 2.21e+01  -1.0 9.20e+00    -  9.47e-01 6.43e-01F  1
  52  4.0904825e+02 1.64e-02 3.90e+01  -1.0 1.02e+01    -  1.00e+00 7.32e-01F  1
  53  4.0590746e+02 9.07e-04 2.62e+01  -1.0 2.03e+00    -  9.66e-01 1.00e+00f  1
  54  4.0529136e+02 2.18e-02 2.08e+01  -1.7 4.18e+00    -  1.00e+00 8.46e-01f  1
  55  4.1308191e+02 2.34e-02 1.34e+01   0.1 1.73e+01    -  5.67e-01 2.25e-01f  2
  56  4.0525564e+02 2.51e-02 3.03e+01  -0.3 7.92e+00    -  1.00e+00 1.00e+00F  1
  57  4.0500304e+02 1.94e-03 1.35e+01  -0.2 1.67e+00    -  9.47e-01 1.00e+00f  1
  58  4.0135256e+02 5.30e-03 2.00e+01  -1.0 2.20e+00    -  1.00e+00 1.00e+00f  1
  59  3.9965183e+02 4.37e-03 1.51e+01  -1.4 2.35e+00    -  1.00e+00 8.47e-01f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
```

```
  60  3.9911968e+02 9.40e-03 1.73e+01  -2.1 3.50e+00    -  1.00e+00 1.00e+00f  1
  61  3.9817039e+02 1.10e-03 1.02e+01  -1.7 1.65e+00    -  1.00e+00 1.00e+00f  1
  62  3.9807105e+02 2.60e-03 9.51e+00  -1.5 1.25e+00    -  1.00e+00 1.00e+00f  1
  63  3.9844528e+02 1.93e-02 2.49e+01  -0.2 1.54e+02    -  1.00e+00 2.34e-02f  2
  64  4.0408697e+02 8.40e-02 3.42e+01  -0.3 7.50e+00    -  1.00e+00 1.00e+00f  1
  65  3.9884197e+02 1.15e-03 2.26e+01  -0.3 1.81e+00    -  1.00e+00 1.00e+00f  1
  66  3.9798411e+02 1.84e-03 1.69e+01  -0.3 1.19e+00    -  1.00e+00 1.00e+00f  1
  67  3.9572215e+02 1.14e-03 1.99e+01  -1.2 6.77e-01    -  9.94e-01 1.00e+00f  1
  68  3.9483194e+02 4.53e-04 1.45e+01  -2.5 5.18e-01    -  1.00e+00 1.00e+00f  1
  69  3.9492494e+02 6.98e-04 4.69e+00  -1.3 1.11e+00    -  9.93e-01 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  70  3.9459402e+02 1.03e-03 1.04e+01  -1.5 1.83e+00    -  1.00e+00 1.00e+00F  1
  71  3.9896254e+02 5.38e-04 1.05e+01  -0.8 3.95e+00    -  9.21e-01 1.00e+00H  1
  72  3.9583460e+02 1.04e-02 1.78e+01  -0.9 5.23e+00    -  9.37e-01 1.00e+00f  1
  73  3.9493705e+02 2.23e-03 1.42e+01  -0.9 1.26e+00    -  1.00e+00 1.00e+00f  1
  74  3.9436931e+02 8.99e-04 1.56e+01  -0.9 3.03e+00    -  1.00e+00 1.00e+00F  1
  75  3.9375967e+02 7.89e-04 2.50e+00  -1.5 9.34e-01    -  9.99e-01 1.00e+00f  1
  76  3.9364432e+02 8.52e-04 4.91e-01  -1.8 8.50e-01    -  9.93e-01 1.00e+00f  1
  77  3.9347350e+02 9.10e-05 5.29e-01  -2.9 2.42e-01    -  1.00e+00 1.00e+00f  1
  78  3.9345205e+02 3.32e-05 2.19e-01  -4.3 1.52e-01    -  1.00e+00 1.00e+00h  1
  79  3.9344876e+02 4.31e-06 1.26e-01  -5.0 4.47e-02    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  80  3.9344836e+02 8.25e-07 1.87e-02  -6.7 2.38e-02    -  1.00e+00 1.00e+00h  1
  81  3.9344834e+02 6.83e-09 3.05e-04  -8.4 1.81e-03    -  1.00e+00 9.98e-01h  1
  82  3.9344834e+02 1.37e-10 4.87e-05 -11.0 4.65e-04    -  1.00e+00 1.00e+00h  1
  83  3.9344834e+02 9.14e-12 7.54e-06 -11.0 1.14e-04    -  1.00e+00 1.00e+00h  1
  84  3.9344834e+02 1.44e-12 2.42e-06 -11.0 3.15e-05    -  1.00e+00 1.00e+00h  1
  85  3.9344834e+02 1.78e-15 2.88e-07 -11.0 5.07e-06    -  1.00e+00 1.00e+00h  1

Number of Iterations....: 85

                                   (scaled)                 (unscaled)
Objective...............:   3.9344833576223050e+02    3.9344833576223050e+02
Dual infeasibility......:   2.8763373595144896e-07    2.8763373595144896e-07
Constraint violation....:   1.7763568394002505e-15    1.7763568394002505e-15
Variable bound violation:   9.9997231828297117e-08    9.9997231828297117e-08
Complementarity.........:   1.0000651152441408e-11    1.0000651152441408e-11
Overall NLP error.......:   2.8763373595144896e-07    2.8763373595144896e-07
```

```
Number of objective function evaluations             = 167
Number of objective gradient evaluations             = 86
Number of equality constraint evaluations            = 167
Number of inequality constraint evaluations          = 0
Number of equality constraint Jacobian evaluations   = 86
Number of inequality constraint Jacobian evaluations = 0
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 3.941

EXIT: Optimal Solution Found.
```






```
[ Info: Listening on: 127.0.0.1:8703, thread id: 1
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browse
r:
└ http://127.0.0.1:8703
```

```
Test Summary:    | Pass  Total   Time
cartpole swingup |    2      2  29.1s
```

Out[ ]:
```
Test.DefaultTestSet("cartpole swingup", Any[], 2, false, false, true, 1.70923715475621e9, 1.709237183856975e9, false)
```

Part C: Track DIRCOL Solution (5 pts)

Now, similar to HW2 Q2 Part C, we are taking a solution X and U from DIRCOL, and we are going to track the trajectory with TVLQR to account for model mismatch. While we used hermite-simpson integration for the dynamics constraints in DIRCOL, we are going to use RK4 for this simulation. Remember to clamp your control to be within the control bounds.

In [ ]:
```
function rk4(params::NamedTuple, x::Vector,u,dt::Float64)
    # vanilla RK4
    k1 = dt*dynamics(params, x, u)
    k2 = dt*dynamics(params, x + k1/2, u)
    k3 = dt*dynamics(params, x + k2/2, u)
    k4 = dt*dynamics(params, x + k3, u)
    x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end

@testset "track cartpole swingup with TVLQR" begin

    X_dircol, U_dircol, t_vec, params_dircol = solve_cartpole_swingup(verbose = false)

    N = length(X_dircol)
    dt = params_dircol.dt
    x0 = X_dircol[1]

    # TODO: use TVLQR to generate K's

    # use this for TVLQR tracking cost
    Q = diagm([1,1,.05,.1])
    Qf = 100*Q
    R = 0.01*diagm(ones(1))

    # initialize K
    K = [zeros(1,4) for i = 1:(N-1)]
    P = Qf
    for i = N:-1:2
        Ac = ForwardDiff.jacobian(_x -> dynamics(params_dircol, _x, U_dircol[i-1]), X_dircol[i-1])
        Bc = ForwardDiff.jacobian(_u -> dynamics(params_dircol, X_dircol[i-1], _u), U_dircol[i-1])
        Z = [Ac Bc; zeros(1,4) zeros(1,1)]
        Zexp = exp(Z*dt)
        A = Zexp[1:4,1:4]
        B = Zexp[1:4,5:5]
        K[i-1] = (R + B'*P*B)\(B'*P*A)
        P = Q + A'*P*A - A'*P*B*K[i-1]
```

```julia
    end

    # simulation
    Xsim = [zeros(4) for i = 1:N]
    Usim = [zeros(1) for i = 1:(N-1)]
    Xsim[1] = 1*x0

    # here are the real parameters (different than the one we used for DIRCOL)
    # this model mismatch is what's going to require the TVLQR controller to track
    # the trajectory successfully.
    params_real = (mc = 1.05, mp = 0.21, l = 0.48)

    # TODO: simulate closed loop system with both feedforward and feedback control
    # feedforward - the U_dircol controls that we solved for using dircol
    # feedback - the TVLQR controls
    for i = 1:(N-1)

        # add controller and simulation step
        Usim[i] = U_dircol[i] - K[i]*(Xsim[i] - X_dircol[i])
        Usim[i] = clamp.(Usim[i], -10, 10)
        Xsim[i+1] = rk4(params_real, Xsim[i], Usim[i], dt)

    end


    # -----------------testing----------------------
    xn = Xsim[N]
    @test norm(xn)>0
    @test 1e-6<norm(xn - X_dircol[end])<.8
    @test abs(abs(rad2deg(xn[2])) - 180) < 5 # within 5 degrees
    @test maximum(norm.(Usim,Inf)) <= (10 + 1e-3)

    # -----------------plotting----------------------
    Xm = hcat(Xsim...)
    Xbarm = hcat(X_dircol...)
    plot(t_vec,Xbarm',ls=:dash, label = "",lc = [:red :green :blue :black])
    display(plot!(t_vec,Xm',title = "Cartpole TVLQR (-- is reference)",
                 xlabel = "time (s)", ylabel = "x",
                 label = ["p" "θ" "ṗ" "θ̇"],lc = [:red :green :blue :black]))
```

```
    Um = hcat(Usim...)
    Ubarm = hcat(U_dircol...)
    plot(t_vec[1:end-1],Ubarm',ls=:dash,lc = :blue, label = "")
    display(plot!(t_vec[1:end-1],Um',title = "Cartpole TVLQR (-- is reference)",
                 xlabel = "time (s)", ylabel = "u",lc = :blue, label = ""))

    # ----------------animate-----------------------
    display(animate_cartpole(Xsim, 0.05))

end
```






```
[ Info: Listening on: 127.0.0.1:8704, thread id: 1
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browse
r:
└ http://127.0.0.1:8704
```

```
Test Summary:                  | Pass  Total  Time
track cartpole swingup with TVLQR |    4      4  7.3s
```

Out[ ]:

```
Test.DefaultTestSet("track cartpole swingup with TVLQR", Any[], 4, false, false, true, 1.70923718388233e9, 1.709237191138534e9, false)
```